

Sharp GP2D12 Analog Distance Sensor (#605-00003)

General Description

The Sharp GP2D12 is an analog distance sensor that uses infrared to detect an object between 10 cm and 80 cm away. The GP2D12 provides a non-linear voltage output in relation to the distance an object is from the sensor and interfaces easily using any analog to digital converter.


Features

- High immunity to ambient light and color of object
- No external control circuitry required
- Sensor includes convenient mounting holes
- Compatible with all BASIC Stamp® and SX microcontrollers

Application Ideas

- Robot range finder
- Halloween prop activation

Quick Start Circuit




Connecting and Testing

Connect the GP2D12 to your analog to digital converter as shown in the circuit on the previous page. The potentiometer connected to the Vref pin on the ADC0831 is being used as a voltage divider to set the reference voltage to 2.55 volts. On the ADC0831 this will give a value of 0 to 255 for an input voltage of 0 to 2.55 volts. This gives us a resolution of 0.01 volts per step from the ADC. If you are using a different analog to digital converter, you may want to adjust the potentiometer to get the best results from your particular ADC.

Calibration

Because the output of the GP2D12 is not linear, we need a way to determine what distances correspond to what voltages. One way of calibrating your sensor is by measuring the voltage output of the GP2D12 at given fixed distances, in centimeters, as shown in the chart below. Once you have this information you can plug these numbers into the EEPROM DATA statements in the program. The table of data is used by a routine in the program to calculate the distances, which are then displayed on the Debug Terminal, along with the voltage output from the sensor.

GP2D12 (voltage vs distance)



Sensitivity

The usable range of the GP2D12 is between 10 cm and 80 cm. The readings for objects closer than 10 cm are unstable and therefore not usable.

Resources and Downloads

Check out the Sharp GP2D12 Analog Distance Sensor product page for example programs, the manufacturer datasheet and more:

http://www.parallax.com/detail.asp?product_id=605-00003



Specifications

Symbol	Quantity	Minimum	Typical	Maximum	Units
Vcc	Supply Voltage †	4.5	5.0	5.5	V
Topr	Operating Temperature †	-10	-	+60	°C
Tstg	Storage Temperature †	-40	-	+70	°C
ΔL	Distance Measuring Range †	10	-	80	cm
Vo	Output Terminal Voltage (L=80 cm) †	0.25	0.4	0.55	V
ΔVo	Output change at L=80 cm to 10 cm †	1.75	2.0	2.25	V
Icc	Average Dissipation Current (L=80cm) †	-	33	50	mA

† data obtained from Sharp's GP2D12 datasheet

Pin Definitions and Ratings

Pin	Name	Function
1	Vo	Voltage Output
2	GND	Ground
3	Vcc	Supply Voltage



*All units in millimeters (mm)

Source Code

BASIC Stamp[®] 2 Program

This sample program reads the GP2D12 and displays the distance on the Debug Terminal. Be sure your GP2D12 is connected as in the Quick Start Circuit. This program will run on the BS2, BS2e, BS2sx, BS2p24, BS2p40, BS2pe and BS2px.

```
' =====
' File..... GP2D12 Demo.bs2
' Purpose... Demonstrate GP2D12
' Author.... Parallax, Inc.
' E-mail.... support@parallax.com
' {$STAMP BS2}
' {$PBASIC 2.5}
' =====

' -----[ Program Description ]-----

' This program demonstrates reading the distance in centimeters from the
' Sharp GP2D12 Analog Distance Sensor.

' -----[ I/O Definitions ]-----

Adc0831      PIN      0          ' ADC0831 Chip Select (ADC0831.1)
AdcClock     PIN      1          ' ADC0831 Clock (ADC0831.7)
AdcData      PIN      2          ' ADC0831 Data (ADC0831.6)

' -----[ Constants ]-----

span         CON      5          ' 5 cm Per Data Point

' -----[ Variables ]-----

result       VAR      Byte       ' ADC8031 Result
volts        VAR      Word       ' Volts (0.01 Increments)
cm           VAR      Byte       ' centimeters
index        VAR      Nib        '
test1        VAR      Byte       ' Values For
test2        VAR      Byte       ' Interpolation
slope        VAR      Word       ' mV/cm between test points

' -----[ EEPROM Data ]-----

Vout         DATA    251, 179, 139, 114, 97
              DATA    85, 76, 67, 62, 57
              DATA    53, 50, 48, 46, 43
              DATA    0

' -----[ Initialization ]-----

HIGH Adc0831          ' Disable ADC0831
```

```

' -----[ Program Code ]-----

DO
  GOSUB Read_GP2D12          ' Read Sensor Value
  GOSUB Calculate_Distance   ' Convert Value To cm
  DEBUG HOME, "Distance = ", DEC cm, " cm "
  PAUSE 100
LOOP
END

' -----[ Subroutines ]-----

Read_GP2D12:
  volts = 0                  ' Reset Sensor Value
  FOR index = 0 TO 2        ' Read 3 Times
    LOW Adc0831             ' Enable ADC0831
    SHIFTIN AdcData, AdcClock, MSBPOST, [result\9] ' Read The Voltage
    HIGH Adc0831            ' Disable ADC0831
    volts = volts + result  ' Add The Values
    PAUSE 30
  NEXT
  volts = volts / 3         ' Average The Readings
  RETURN

Calculate_Distance:
  FOR index = 0 TO 15       ' Search DATA Table For Value
    READ (Vout + index), test2 ' Get Value From DATA Table
    IF (test2 <= volts) THEN EXIT ' Found Value
  NEXT
  SELECT index
  CASE 0
    cm = 10                 ' Set To Minimum Distance
  CASE 1 TO 14
    cm = 10 + (5 * index)   ' Calculate Distance
    IF (test2 < volts) THEN ' Estimate Using Interpolation
      READ (Vout + index - 1), test1
      slope = (test1 - test2) * 10 / span ' Calculate Slope
      cm = cm - ((volts - test2) * 10 / slope)
    ENDIF
  CASE 15
    cm = 80                 ' Set To Maximum Distance
  ENDSELECT
  RETURN

```